# Multithread module

QT += core
TEMPLATE = lib
CONFIG += dll

DESTDIR = $$PWD/dist

CONFIG(debug,debug|release) {
    TARGET = ZMultiThread_d
} else {
    TARGET = ZMultiThread
}

include($$PWD/../ZFLibCommon.pri)

DEFINES += ZFLIB_MAKER

HEADERS += \
    $$PWD/ZMultiThreadCtrl.h \
    $$PWD/ZMultiThreadTask.h

SOURCES += \
    $$PWD/ZMultiThreadCtrl.cpp \
    $$PWD/ZMultiThreadTask.cpp

# install
ZMultiThread_1.files = $$PWD/ZMultiThreadTask.h \
    $$PWD/ZMultiThreadCtrl.h
ZMultiThread_1.path =  $$DIST_DIR/ZMultiThread/include
INSTALLS += ZMultiThread_1

ZMultiThread_2.files = $$PWD/ZMultiThread.pri
ZMultiThread_2.path =  $$DIST_DIR/ZMultiThread
INSTALLS += ZMultiThread_2

# install to DIST_DIR
defined(DIST_DIR, var) {
ZMultiThread_3.files = $$PWD/dist/*
ZMultiThread_3.path =  $$DIST_DIR/ZMultiThread/bin
INSTALLS += ZMultiThread_3
}
